aula online - sexta, 20 de novembro

equações às diferenças

1.

```
ex1[k_] := 1 - 2 ex1[k - 1]^2
ex1[0] = 0.86;
```

```
ex1[1]
```

-0.4792

```
ex1[2]
```

0.540735

```
ex1[3]
```

0.415212

```
ex1[4]
```

0.655198

```
ex1[5]
```

0.141431

uma alternativa, calculando de uma só vez, através do comando Table, todos os cinco valores pretendidos

```
Table[ex1[k], {k, 1, 5}]
```

{-0.4792, 0.540735, 0.415212, 0.655198, 0.141431}

2.

```
ex2[k_] := 4 ex2[k - 1] - 2 ex2[k - 1]^2
ex2[0] = 0.2;
```

```
Table[ex2[k], {k, 1, 6}]
```

```
{0.72, 1.8432, 0.578028, 1.64388, 1.17084, 1.94163}
```

3.

```
ex3[k_] := ex3[k - 1] - ex3[k - 2]
ex3[0] = 1;
ex3[1] = 2;
```

```
Table[ex3[k], {k, 1, 8}]
```

```
{2, 1, -1, -2, -1, 1, 2, 1}
```

4.

desta vez vamos usar a função f(x) que descreve a equação às diferenças de primeira ordem x_k=f(x_{k-1})

```
f04[x_] = Sin[1.6 x];
```

assim, podemos usar o comando NestList, que vai iterar a função indicada, a partir de um determinado ponto

```
NestList[f04, -0.185, 8]
```

```
{-0.185, -0.291697, -0.449955, -0.65933,
 -0.869865, -0.98402, -0.999993, -0.999574, -0.999593}
```

em termos da função f(x), as soluções de tipo constante x_k=x_{k-1} surgem como f(x)=x. assim sendo, estas soluções vão corresponder às intersecções dos gráficos de f(x) e da recta y=x

```
Plot[{x, f04[x]}, {x, -2.7, 2.7}, PlotRange → {-1.6, 1.6},
 AspectRatio → Automatic, Frame → True, ImageSize → 500]
```



pelo gráfico acima, podemos concluir que o sistema dinâmico descrito pela equação às diferenças dada no enunciado admite três soluções de tipo constante (sendo uma delas facilmente identificável como x=0)

nota. não sendo f(x) = x uma equação polinomial, não pode ser resolvida com o comando Solve. no entanto, podemos calcular as outras duas soluções de tipo constante (na realidade apenas uma delas, pela simetria do problema) usando o comando FindRoot. contudo, este comando necessita que seja dada uma aproximação para a solução pretendida (estaremos a usar o método de newton) coisa que conseguimos facilmente pelo gráfico acima.

```
FindRoot[f04[x] == x, {x, -1}]
```

```
{x → -0.999592}
```

em termos da função f(x), as soluções periódicas, de período 2, $x_k = x_{k-2}$ da equação às diferenças, surgem como f(f(x)) = x. assim sendo, estas soluções vão corresponder às intersecções dos gráficos de f(f(x)) e y=x (não esquecer que é necessário que x_{k-1} seja diferente de x_{k-2})

```
Plot[{x, f04[x], f04[f04[x]]}, {x, -2.7, 2.7}, PlotRange → {-1.6, 1.6},
 AspectRatio → Automatic, Frame → True, ImageSize → 600]
```



pelo gráfico acima, podemos concluir que o sistema dinâmico descrito pela equação às diferenças dada no enunciado não admite qualquer ciclo de período dois (isto é, podemos concluir que a equação às diferenças não tem qualquer solução periódica de período 2)

5.

vamos definir a função f(x) que descreve o sistema dinâmico, sem indicar o intervalo [0,1]. contudo, teremos que ter muita atenção aos resultados que vamos obter, uma vez que o sistema dinâmico só está definido para valores no intervalo indicado.

```
f05[x_] = 1.2 x + 2.8 x^2 - 4 x^3;
```

```
NestList[f05, 0.95, 20]
```

```
{0.95, 0.2375, 0.389352, 0.655592, 0.863057, 0.549846, 0.8414,
 0.609262, 0.865842, 0.541694, 0.835842, 0.623397, 0.867155, 0.537811,
 0.83302, 0.630403, 0.867117, 0.537924, 0.833104, 0.630196, 0.867125}
```

o comando ListPlot permite-nos apresentar graficamente os primeiros elementos da órbita do ponto x0=0.95 (a utilização da opção DataRange é para fazer corresponder o valor inicial 0.95 à abcissa k=0)

```
ListPlot[NestList[f05, 0.95, 20],
 DataRange → {0, 20}, Joined → True, Mesh → All, Frame → True]
```



```
Plot[{x, f05[x]}, {x, 0, 1}, AspectRatio → Automatic, Frame → True, ImageSize → 400]
```



o gráfico acima, desenhado apenas para valores de x no intervalo [0,1], permite-nos concluir que o sistema dinâmico admite duas soluções de tipo constante (uma delas facilmente identificável como x = 0)

```
Solve[f05[x] == x]
```

{{x → -0.0653312}, {x → 0.}, {x → 0.765331}}

assim sendo, podemos concluir que o sistema dinâmico admite duas soluções de tipo constante x = 0 & x = 0.765331

```
Plot[{x, f05[x], f05[f05[x]]}, {x, 0, 1},
 AspectRatio → Automatic, Frame → True, ImageSize → 400]
```



o gráfico acima, desenhado apenas para valores de x no intervalo [0,1], permite-nos concluir que o sistema dinâmico admite um ciclo de período dois (existem duas intersecções de f(f(x)) com a recta y=x, em pontos onde f(x) e a recta y = x não se intersectam)

```
Solve[f05[f05[x]] == x]
```

{{x → -0.591325}, {x → -0.266767 - 0.192267 i}, {x → -0.266767 + 0.192267 i},
 {x → -0.0653312}, {x → 0.}, {x → 0.573635}, {x → 0.765331}, {x → 0.854687}, {x → 1.09654}}

assim sendo, podemos concluir que o sistema dinâmico admite um ciclo de período 2, dado pelo par {0.573635, 0.854687}

6.

```
f06[x_] = -1.82 - 0.74 x - 0.16 x^2
```

$-1.82 - 0.74\,x - 0.16\,x^2$

```
Plot[{x, f06[x]}, {x, -14, 3}, PlotRange → {-14, 4}, Frame → True]
```



sendo f(x) uma parábola, podemos afirmar que não existem quaisquer outras intersecções de f(x) com a recta y = x

```
Solve[f06[x] == x]
```

{{x → -9.70264}, {x → -1.17236}}

o sistema dinâmico admite duas soluções de tipo constante/pontos fixos x = -9.70264 & x = -1.17236

```
Plot[{x, f06[x], f06[f06[x]]}, {x, -14, 3},
 PlotRange → {-14, 4}, Frame → True, ImageSize → 500]
```



sendo f(f(x)) um polinómio de grau quatro, e uma vez que no domínio do gráfico acima é possível identificar os pontos correspondentes a três extremos de f(f(x)), podemos afirmar com segurança que não existem mais intersecções de f(f(x)) com a recta y = x

logo, o sistema dinâmico não admite qualquer ciclo de período 2. vamos confirmar?

```
Solve[f06[f06[x]] == x]
```

{{x → -9.70264}, {x → -1.17236}, {x → 0.8125 - 4.56849 ⅈ}, {x → 0.8125 + 4.56849 ⅈ}}

as duas primeiras soluções são as duas soluções de tipo constante/pontos fixos do sistema.
as restantes são números complexos, logo não pertencem ao domínio de f(x)

7.

```
f07[x_] = 1.982 - 2.306 x - 0.708 x^2
```

$1.982 - 2.306\,x - 0.708\,x^2$

```
Plot[{x, f07[x]}, {x, -7, 5}, PlotRange → {-7, 5}, AspectRatio → Automatic, Frame → True]
```



sendo f(x) uma parábola, podemos afirmar que não existem quaisquer outras intersecções de f(x) com a recta y = x

```
Solve[f07[x] == x]
```

{{x → −5.20711}, {x → 0.537618}}

o sistema dinâmico admite duas soluções de tipo constante/pontos fixos x = -5.20711 & x = 0.537618

```
Plot[{x, f07[x], f07[f07[x]]}, {x, -6, 3},
 PlotRange → {-18, 6}, Frame → True, ImageSize → 500]
```



sendo f(f(x)) um polinómio de grau quatro, podemos afirmar com toda a segurança que não existem mais intersecções de f(f(x)) com a recta y = x

logo, o sistema dinâmico admite um ciclo de período 2

```
Solve[f07[f07[x]] == x]
```

{{x → -5.20711}, {x → -3.42342}, {x → 0.537618}, {x → 1.57879}}

das quatro soluções obtidas, facilmente se reconhece que duas delas são exactamente os pontos fixos identificados anteriormente. logo, podemos afirmar que o sistema dinâmico admite um ciclo de período 2, dado pelo par {-3.42342, 1.57879}

8.

```
f08[x_] = -1.284 - 3.128 x - 2.671 x^2 - 0.722 x^3
```

$-1.284 - 3.128\,x - 2.671\,x^2 - 0.722\,x^3$

```
Plot[{x, f08[x]}, {x, -3, 1}, PlotRange → {-6, 4}, Frame → True]
```



sendo f(x) uma cúbica, podemos afirmar que não existem quaisquer outras intersecções de f(x) com a recta y = x

```
Solve[f08[x] == x]
```

{{x → −1.64673 − 1.29175 ⅈ}, {x → −1.64673 + 1.29175 ⅈ}, {x → −0.405996}}

das três soluções obtidas, duas delas são números complexos. logo, podemos concluir que o sistema dinâmico admite uma única solução de tipo constante/ponto fixo x = -0.405996

```
Plot[{x, f08[x], f08[f08[x]]}, {x, -3, 1},
 PlotRange → {-3, 1}, Frame → True, ImageSize → 500]
```



pelo gráfico acima (o domínio do gráfico foi escolhido depois de termos percebido que não existem quaisquer outras intersecções de f(f(x)) com a recta y = x à esquerda e à direita dos valores escolhidos) podemos concluir que o sistema dinâmico admite dois ciclos de período 2

```
Solve[f08[f08[x]] == x]
```

```
{{x → -2.20553 - 1.10543 ⅈ}, {x → -2.20553 + 1.10543 ⅈ},
 {x → -2.12026}, {x → -1.64673 - 1.29175 ⅈ}, {x → -1.64673 + 1.29175 ⅈ},
 {x → -0.98568}, {x → -0.405996}, {x → -0.104418}, {x → 0.222521}}
```

das nove soluções encontradas, quatro delas são números complexos, que devemos rejeitar, e duas são os pontos fixos encontrados anteriormente. logo, o que nos interessa são as restantes quatro soluções de f(f(x)) = x (o que corresponde exactamente à conclusão retirada a partir do gráfico acima)

agora teremos que ver que pares de pontos correspondem aos ciclos de período 2 do sistema dinâmico

```
f08[-0.9856799271135434`]
```

```
-0.104418
```

assim sendo, um deles é dado por {-0.98568, -0.104418}

sendo o outro dado por {-2.12026, 0.222521}

nota. não há necessidade de fazer copy/paste do valor da sexta solução obtida anteriormente. podemos transformar a resposta do comando Solve numa lista de soluções e então pedir para que seja calculada a imagem por f do sexto elemento dessa lista

```
solucoes08 = x /. Solve[f08[f08[x]] == x]
```

```
{-2.20553 - 1.10543 ⅈ, -2.20553 + 1.10543 ⅈ, -2.12026, -1.64673 - 1.29175 ⅈ,
 -1.64673 + 1.29175 ⅈ, -0.98568, -0.405996, -0.104418, 0.222521}
```

```
f08[Part[solucoes08, 6]]
```

```
-0.104418
```