sistemas dinâmicos discretos

03 dezembro 2020

01. dinâmicas

soluções de tipo constante - pontos fixos

In[1]:=
```
f01[x_] = 4 x (1 - x)
```
Out[1]= $4(1-x)x$

In[2]:=
```
Plot[{x, f01[x]}, {x, 0, 1}, AspectRatio → 0.85, Frame → True]
```
Out[2]=



In[3]:=
```
Solve[f01[x] == x]
```
Out[3]= $\left\{\{x \to 0\}, \left\{x \to \frac{3}{4}\right\}\right\}$

soluções periódicas - ciclos

In[4]:=
```
Plot[{x, f01[x], f01[f01[x]]}, {x, 0, 1}, AspectRatio → 0.85, Frame → True]
```

Out[4]=



In[5]:=
```
Solve[f01[f01[x]] == x]
```

Out[5]= $\left\{\{x \to 0\}, \left\{x \to \frac{3}{4}\right\}, \left\{x \to \frac{1}{8}\left(5 - \sqrt{5}\right)\right\}, \left\{x \to \frac{1}{8}\left(5 + \sqrt{5}\right)\right\}\right\}$

02. estabilidade: pontos fixos

ponto fixo atractivo ⇌ ponto fixo repulsivo

consideremos um sistema dinâmico discreto descrito por uma função diferenciável *f*

seja xo um ponto fixo de *f*

1. se $|f'(xo)| < 1$, então xo é um ponto fixo atractivo de *f*

2. se $|f'(xo)| > 1$, então xo é um ponto fixo repulsivo de *f*

nota. se $|f'(xo)| = 1$, então xo pode ser um ponto fixo atractivo, ou um ponto fixo repulsivo, ou não ser nem atractivo, nem repulsivo, de *f*

In[6]:=
```
f02[x_] = 2.4 x (1 - x)
```

Out[6]= 2.4 (1 - x) x

In[45]:=
```
Solve[f02[x] == x]
fixos02 = x /. Solve[f02[x] == x]
```

Out[45]= {{x → 0.}, {x → 0.583333}}

Out[46]= {0., 0.583333}

In[9]:=
```
Abs[f02'[Part[fixos02, 1]]]
```

Out[9]= 2.4

In[10]:=
```
Abs[f02'[Part[fixos02, 2]]]
```

Out[10]= 0.4

In[11]:=
```
orbita02 = NestList[f02, 0.95, 30];
```

In[12]:=
```
ListPlot[orbita02, PlotRange → {0, 1},
 Joined → True, Mesh → True, Frame → True, DataRange → {0, 31}]
```

Out[12]=



In[13]:=
```
estadoFinal02 = Take[NestList[f02, RandomReal[{0, 1}], 100], -40];
Histogram[estadoFinal02, {0, 1, 0.02}, Frame → True]
```

Out[14]=


03. estabilidade: ciclos de período 2

2-ciclo atractivo ⇌ 2-ciclo repulsivo

consideremos um sistema dinâmico discreto descrito por uma função diferenciável *f*

seja xo um ponto fixo de f^2

1. se $|(f^2)'(xo)| < 1$, então xo é um ponto fixo atractivo de f^2

2. se $|(f^2)'(xo)| > 1$, então xo é um ponto fixo repulsivo de f^2

nota. se $|(f^2)'(xo)| = 1$, então xo pode ser um ponto fixo atractivo, ou um ponto fixo repulsivo, ou não ser nem atractivo, nem repulsivo de f^2

In[15]:=
```
D[f[f[x]], x]
```

Out[15]= f′[x] f′[f[x]]

In[16]:=
```
f03[x_] = 3.4 x (1 - x)
```

Out[16]= 3.4 (1 - x) x

In[49]:=
```
x /. Solve[f03[x] == x]
x /. Solve[f03[f03[x]] == x]
```

Out[49]= {0., 0.705882}

Out[50]= {0., 0.451963, 0.705882, 0.842154}

In[19]:=
```
ciclo2 = Part[x /. Solve[f03[f03[x]] == x], {2, 4}]
```
Out[19]= {0.451963, 0.842154}

In[20]:=
```
Abs[f03'[Part[ciclo2, 1]] * f03'[Part[ciclo2, 2]]]
```
Out[20]= 0.76

In[71]:=
```
orbita03 = NestList[f03, 0.95, 40];
ListPlot[orbita03, PlotRange → {0, 1}, Joined → True, Mesh → True,
 Frame → True, DataRange → {0, 31}, AspectRatio → 0.3, ImageSize → 600]
```
Out[72]=



In[23]:=
```
estadoFinal03 = Take[NestList[f03, RandomReal[{0, 1}], 100], -40];
Histogram[estadoFinal03, {0, 1, 0.02}, Frame → True]
```
Out[24]=



experimentalmente, chegamos à conclusão que o sistema dinâmico discreto descrito por f tem um único atractor: um ciclo de período 2

04. exemplo

In[25]:=
```
f04[x_] = 3.54 x (1 - x)
```

Out[25]= 3.54 (1 - x) x

In[69]:=
```
orbita04 = NestList[f04, 0.95, 50];
ListPlot[orbita04, PlotRange → {0, 1}, Joined → True, Mesh → True,
 Frame → True, DataRange → {0, 31}, AspectRatio → 0.3, ImageSize → 600]
```

Out[70]=



In[51]:=
```
estadoFinal04 = Take[NestList[f04, RandomReal[{0, 1}], 10 000], -9000];
Histogram[estadoFinal04, {0, 1, 0.02}, Frame → True]
```

Out[52]=



experimentalmente, chegamos à conclusão que o sistema dinâmico discreto descrito por f tem um único atractor: um ciclo de período 4

05. exemplo

In[30]:=
```
f05[x_] = 3.84 x (1 - x)
```

Out[30]= 3.84 (1 - x) x

In[65]:=
```
orbita05 = NestList[f05, 0.95, 50];
ListPlot[orbita05, PlotRange → {0, 1}, Joined → True, Mesh → True,
 Frame → True, DataRange → {0, 31}, AspectRatio → 0.3, ImageSize → 600]
```

Out[66]=



In[33]:=
```
estadoFinal05 = Take[NestList[f05, RandomReal[{0, 1}], 10 000], -210];
Histogram[estadoFinal05, {0, 1, 0.02}, Frame → True]
```

Out[34]=



experimentalmente, chegamos à conclusão que o sistema dinâmico discreto descrito por f tem um único atractor: um ciclo de período 3

06 exemplo

In[35]:=
```
f06[x_] = 3.89 x (1 - x)
```

Out[35]= 3.89 (1 - x) x

In[59]:=
```
orbita06 = NestList[f06, 0.95, 100];
ListPlot[orbita06, PlotRange → {0, 1}, AspectRatio → 0.25, ImageSize → 600,
 Joined → True, Mesh → True, Frame → True, DataRange → {0, 101}]
```

Out[60]=



In[38]:=
```
estadoFinal06 = Take[NestList[f06, RandomReal[{0, 1}], 60 000], -50 000];
Histogram[estadoFinal06, {0, 1, 0.02}, Frame → True]
```

Out[39]=



experimentalmente, chegamos à conclusão que o sistema dinâmico discreto descrito por *f* tem um único atractor

mas este atractor não corresponde a uma dinâmica de tipo constante ou periódica!

07. exemplo

In[40]:=
```
f07[x_] = 4. x (1 - x)
```

Out[40]= 4. (1 - x) x

In[57]:=
```
orbita07 = Take[NestList[f07, RandomReal[{0, 1}], 1000], -100];
ListPlot[orbita07, PlotRange → {0, 1}, AspectRatio → 0.25, ImageSize → 600,
 Joined → True, Mesh → True, Frame → True, DataRange → {1000, 1100}]
```

Out[58]=



In[41]:=
```
estadoFinal07 = NestList[f07, RandomReal[{0, 1}], 60 000];
Histogram[estadoFinal07, {0, 1, 0.02}, Frame → True]
```

Out[42]=



experimentalmente, chegamos à conclusão que o sistema dinâmico discreto descrito por f tem um único atractor

não sendo periódico, vamos dizer que estamos perante um atractor aperiódico